Data Movement Tips and Tricks

David Appelhans

IBM Research,

ORNL SUMMIT workshop
March 7, 2018
INTRODUCTION

• This is an early version of 50 min talk for GTC 2018–For more complete details, please see that talk.

• Importance of pinned memory. (Interoperability, CUDA+OpenMP+OpenACC)

• Zero-copy tricks. (Interoperability, CUDA+OpenMP)

• Dealing with nested data structures. (Efficiency, CUDA Fortran)
**Motivation: Why you should pin your memory**

Pageable vs Pinned HtoD Bandwidth Impact

Dual socket P9 + 6 Volta GPUs

- OpenACC
- OpenMP
- CUDA

Example code runnable on SUMMIT is available at [https://github.com/dappelha/gpu-tips/tree/master/pinned-memory](https://github.com/dappelha/gpu-tips/tree/master/pinned-memory)
Pinned Memory Option 1:

Use CUDA Fortran\(^1\) pinned attribute to pin at allocation time,

\begin{verbatim}
real (kind=8), pinned, allocatable :: p_A(:)
allocate ( p_A(N) )
!$omp target data map(alloc:p_A)
do i=1,samples
   !$omp target update to(p_A)
   ...
enddo
\end{verbatim}

Can also check success of pinning:

\begin{verbatim}
logical :: pstat
allocate ( p_A(N), pinned=pstat)
if (. not. pstat ) print *, "ERROR: p_A was not pinned"
\end{verbatim}

\(^1\)PGI and XLF compilers both support CUDA Fortran, so the pinned attribute can easily be combined with directives.
Pinned Memory Option 2:

Pin already allocated memory,\(^2\)

```fortran
use, intrinsic :: iso_c_binding
use cudafor
real, pointer, contiguous :: phi (:,:)
allocate ( phi(dim1, dim2) ) ! phi can also be pointer passed from C++
istat = cudaHostRegister(C_LOC(phi(1,1)), sizeof(phi), cudaHostRegisterMapped)

!$acc enter data create (phi)
do i=1,samples
   !$acc update self (phi)
   ...
enddo
```

Warning: pinning memory is very slow. Memory should only be pinned if it is going to be used for many transfers.

\(^2\)This technique is especially useful if the memory was allocated outside the developer's control (for example in a C++ calling routine).
You must use the flag `-ta=tesla:pinned` in order for OpenACC to benefit from pinned memory. Without this flag you will see the same bandwidth as pageable memory. There are two ways to use the flag:

1. **Compiling** with the flag `-ta=tesla:pinned` forces all memory to be pinned memory. If you have many arrays that are not used by the GPU, you will unnecessarily be paying the expensive cost to pin them.

2. **Linking** the final executable with `-ta=tesla:pinned` causes the OpenACC runtime to check if an array is already pinned. This gives fine grain user control to manually pin arrays as shown in the previous examples.
OpenACC Interoperability Warning 2

The OpenACC runtime uses a memory pool on the device to save from repeated allocation/deallocation of device memory. This is good for exclusive OpenACC codes but it can cause trouble when mixing programming models:

```fortran
integer :: N = 8*gigabyte
real (kind=8), allocatable :: A(:)
real (kind=8), device, allocatable :: d_A(:)
allocate ( A(N) )
!$acc enter data create (A)
!$acc exit data delete (A) ! <---not truly free’d unless PGI_ACC_MEM_MANAGE=0
allocate ( d_A(N) ) ! <------ can then run out of device memory
```

To disable this optimization, set the environment flag `PGI_ACC_MEM_MANAGE=0` and the runtime will free the data at the exit data.
USES OF ZERO COPY

Zero copy refers to accessing host resident pinned memory directly from a GPU without having to copy the data to the device beforehand (i.e. there are zero device copies).

- Quick overlap of data movement and kernel compute (mostly bad idea because it hides compute/movement distinction, and unified/managed memory now serves this purpose)
- Reads or writes of non unit stride arrays.
  - Example 1: Large arrays where only small percent of data is accessed in random pattern.
  - Example 2: All data is accessed, but read/write pattern is strided/not coalesced.
  - Example 3: Efficiently populating elements of a structure, avoiding the overhead of many copy api calls by using GPU threads to fetch data directly.
CUDA Zero Copy Setup

To set up zero copy of a basic array in Fortran, use a CUDA API to get a device pointer that points to the pinned host array, and then associate a fortran array with that C device pointer, specifying the Fortran array attributes.

```fortran
use iso_c_binding ! provides c_f_pointer and C_LOC
! zero copy pointers for psib

! sets up zero copy of psib needed for snrefelctD on device.
istat = cudaHostGetDevicePointer(d_psib_p, C_LOC(psib(1,1,1)), 0)
! Translate that C pointer to the fortran array with given dimensions

! sets up zero copy of psib needed for snrefelctD on device.
istat = cudaHostGetDevicePointer(d_psib_p, C_LOC(psib(1,1,1)), 0)
```

**OPENMP ZERO COPY EXAMPLE**

Only requires CUDA pinned array and OpenMP is_device_ptr clause.

```fortran
real (kind=8), pinned, allocatable :: A (:,:), At (:,:)
...
allocate ( A(nx,ny), At(ny,nx) )

! Transpose in the typical way:
!$omp target enter data map(alloc:A,At)
call transpose (A,At,nx,ny)
!$omp target update from(At)
!$omp target exit data map(delete:At)

! Ensure device has finished for accurate benchmarking
ierr = cudaDeviceSynchronize()

! Transpose using zero copy for At.
! At is no longer mapped—-is_device_ptr(At) will
! allow addressing host pinned memory (zero copy)
call transpose_zero_copy (A,At,nx,ny)
```

continued on next slide
OpenMP Zero Copy Example Continued

```fortran
subroutine transpose_zero_copy(A, At, nx, ny)
  ! example of strided writes to an array that lives on the host
  ! variable declarations
  implicit none
  real (kind=8), intent (in) :: A (:,:)
  real (kind=8), intent (out) :: At (:,:)
  integer , intent (in) :: nx, ny
  integer :: i, j
  !$omp target teams distribute parallel do is_device_ptr (At)
  do j=1,ny
    do i=1,nx
      At(j,i) = A(i,j)
    enddo
  enddo
  return
end subroutine transpose_zero_copy
```

D. Appelhans

Data Movement Tips and Tricks
**OpenMP Zero Copy Transpose**

Table: Power8 + P100 results of doing a traditional matrix transpose and then copying back from GPU vs doing the transpose directly into pinned host memory.

<table>
<thead>
<tr>
<th>Method</th>
<th>Time (ms)</th>
<th>Total Time</th>
</tr>
</thead>
<tbody>
<tr>
<td>transpose + memcpy</td>
<td>57.3 + 63.1</td>
<td>120.4</td>
</tr>
<tr>
<td>zero-copy-transpose</td>
<td>65.4</td>
<td>65.4</td>
</tr>
</tbody>
</table>
**Efficiently Populating Nested Structures**

A naive implementation for getting data structures populated on the GPU usually looks like this:

```plaintext
1 % ! allocate GPUelement
2 % do id=1, Nelements
3 %    call construct_GPUelements(GPUelement,id,Nnodes)
4 % enddo
5 %
6 ! still need to populate the values from the host version of the data structure:
7 do id=1, Nelements
8    GPUelement(id)%Nnodes = element(id)%Nnodes ! implicit cudaMemcpy
9    GPUelement(id)%x = element(id)%x       ! implicit cudaMemcpy
10   GPUelement(id)%y = element(id)%y       ! implicit cudaMemcpy
11   GPUelement(id)%val = element(id)%val    ! implicit cudaMemcpy
12 enddo
13 %
14 % do id=1,Nelements
15 %    call destruct_GPUelements(GPUelement,id)
16 % enddo
```

This becomes incredibly slow when `Nelements` is large.
Efficiently Populating Nested Structures

There are two ways to fix the naive approach,

1. loop over cudaMemcpyAsync calls instead of the numerous blocking calls done above,

2. pull the data from the GPU by populating the arrays from within a device kernel using zero copy.
**Push Data by Looping Over cudaMemcpyAsync**

CUDA Fortran code would look like this:

```fortran
integer (cuda_stream_kind) old_stream, streamid
ierr = cudaStreamCreate(streamid)
old_stream = cudaforGetDefaultStream() ! save the current default stream
ierr = cudaforSetDefaultStream(streamid) ! Set the default stream to streamid
do id=1, Nelements
  GPUelement(id)%Nnodes = element(id)%Nnodes ! implicit cudaMemcpyAsync on streamid
  GPUelement(id)%x = element(id)%x ! implicit cudaMemcpyAsync on streamid
  GPUelement(id)%y = element(id)%y ! implicit cudaMemcpyAsync on streamid
  GPUelement(id)%val = element(id)%val ! implicit cudaMemcpyAsync on streamid
endo
ierr = cudaforSetDefaultStream(old_stream) ! restore the original default stream
```

Can do similar in OpenMP 4, with update nowait (no example yet).
**PULLING FROM THE GPU**

Set up a way to reference host structures from the device (zero copy of a structure):

```c
1! to use element on the device, we have to make a device valid copy called d_element:
2  istat = cudaMemcpyAsync(d_element, element, size(element), 0)
3! similarly with d_GPUelement, we need to transfer the device memory addresses held
   in GPUelement
4  istat = cudaMemcpyAsync(d_GPUelement, GPUelement, size(GPUelement), 0)
5! Now we can use these in a CUDA kernel to zero copy
6! the member data from d_element into d_GPUelement
7 call set_elements_kernel <<<blocks,threads>>>(d_GPUelement,d_element,Nelements)
```

- line 4 allows deep references within a GPU kernel, such as d_GPUelement(id)%d_member.
- Can also accomplish this by making GPUelement a managed memory variable.
Launch a kernel to have GPU threads pull data from structures on the host into structures on the device:

```
attributes (global) subroutine set_elements_kernel (GPUelement, element, Nelements)
    implicit none
    ! kernel that uses zero copy to populate the GPUelement structure.
    ! I have dropped the d_ prefix for convenience on these dummy variables
    type (GPUelement_type), device, intent (inout) :: GPUelement(:) ! members are device
    type (element_type), device, intent (in) :: element(:) ! members are pinned host
    integer, value, intent (in) :: Nelements
    integer :: id, Nnodes, node

    do id = blockIdx%x, Nelements, blockDim%x
        Nnodes = element(id)%Nnodes
        GPUelement(id)%Nnodes = Nnodes
        do node = threadIdx%x, Nnodes, blockDim%x
            GPUelement(id)%x(node) = element(id)%x(node)
            GPUelement(id)%y(node) = element(id)%y(node)
            GPUelement(id)%val(node) = element(id)%val(node)
        enddo
    enddo
end subroutine set_elements_kernel
```
**EXAMPLE CODE**

Example code runable on SUMMIT is available at
https://github.com/dappelha/gpu-tips/tree/master/pinned-memory

Includes:

- jsrun submission script example,
- affinity checking script to verify,
- device binding helper script,
- pinned memory comparison example.

Questions: David Appelhans - dappelh@us.ibm.com
 QUESTIONS?

David Appelhans - dappelh@us.ibm.com